Véra Molnar et l'art optique

L'Art optique rassemble des artistes fascinés par ce que l'œil perçoit du mouvement et de la lumière et qui en explorent les propriétés : intensité, rythme, variabilité, cycles... La démarche est bien sûr esthétique (non technique, physique ou métaphysique). C'est un mouvement riche et varié, qui s'exprime par des tableaux, des sculptures mais aussi des assemblages, installations ou montages vidéos. À ses débuts, on parle d'Art cinétique et optique. Dans les années 1960, l'appellation est simplifiée : on parle d'Art optique en France, d'Optique Art ou Op Art chez les anglo-saxons.

Véra Molnar est née en 1924 à Budapest (Hongrie). Depuis 1947, elle vit et travaille à Paris. Elle peut être présentée comme un peintre géométrique : les éléments de base de son travail sont parmi les plus simples, les plus élémentaires : la ligne, le carré, le blanc, le noir, parfois des gris, des rouges, des bleus… A l'exploration de ces formes, elle a consacré des dizaines d'années ; et elle continue aujourd'hui. Elle a participé à tous les débats qui ont animé la naissance de l'art cinétique et permis la création de La Nouvelle Tendance et est devenue à partir de 1968 l'une des pionnières de l'utilisation de l'ordinateur dans la création artistique. Son art, conduit de façon expérimentale, porte sur la forme, sa transformation, son déplacement, sa perception. Son travail s'accompagne d'une intense réflexion théorique sur les moyens de la création et les mécanismes de la vision et trouve de nombreuses correspondances dans tous les travaux conduits en rapport avec les sciences exactes et les mathématiques en particulier.

Le travail qui, chez d'autres, pourrait être systématique voire "machinique", a en réalité pour but de faire surgir l'imprévu, la liberté, l'imaginaire. Les lignes, par exemple, deviennent « extravagantes », comme nous le précisent d'ailleurs les titres de certains tableaux. *(d'après la biographie de Véra Molnar présente sur le site de la galerie ONIRIS).*

Cette œuvre s'inscrit dans une série consacrée aux quadrilatères autour de l'année 1984. Véra Molnar a conçu de nombreuses œuvres, toutes nommées *Structure de quadrilatères*, sur des supports différents : ordinateur, toile, papier ...

Étude géométrique et algorithmique de l'œuvre

L'étude mathématique du tableau est indissociable de sa réalisation avec *GéoTortue*. Les propriétés géométriques se dévoileront au fur et à mesure des problèmes rencontrés avec le logiciel.

Mise en place du problème

L'œuvre peut être divisée en deux parties selon la couleur utilisée (noir et rouge) pour tracer les motifs. Chacune de ce deux parties est composée de groupements de quadrilatères quelconques générés aléatoirement *(voir fig. 1)*.



fig. 1

De plus, le nombre de quarilatères composant chaque groupement suit une logique croissante sur une même ligne puis de ligne en ligne *(voir fig. 2)*.



fig. 2

Il nous faut donc réaliser un quadrilatère quelconque avec *GéoTortue*. Ce travail est décrit précisément dans les commentaires d'une autre œuvre de Véra MOLNAR appelée elle aussi *Structure de quadrilatères*. Ces commentaires sont accessibles sur le site de l'IREM Paris Nord.

La procédure permettant de réaliser un quadrilatère quelconque centré est la suivante :



```
pour quadrilatere l1 l2 l3 a1 a2
//variables intermédiaires
d:= √(l1²+l2²+2*l1*l2*cos(a1))
alpha1:= acos((l1²-l2²-d²)/(-2*l2*d))
alpha2:=180-(a2+alpha1)
beta:=a1-alpha1
gamma:=180-a4-beta
l4:= √(d²+l3²-2*d*l3*cos(alpha2))
a3:= acos((d²-l3²-l4²)/(2*l3*l4))
a4:= 360-a1-a2-a3

//position du centre au sommet
lc
td (180-a4) ; re l4/4 ; tg gamma; re d/4 ; tg beta ; re l1/4
bc

//le tracé
av l1; td a1; av l2 ; td a2 ;av l3 ;td a3 ; av l4 ; td a4

//position du sommet au centre
lc
av l1/4 ; td beta ; av d/4 ; td gamma ;  av l4/4 ; td (180-a4)
bc
fin
```

Création du motif

Le motif est constitué de *n* quadrilatères que nous devons générer aléatoirement. Nous choisissons comme zone de délimitation approximative un carré de 200 de côté et empêchons que le quadrilatère soit trop petit. De plus, pour obtenir un motif qui se rapproche de l'oeuvre, nous devons choisir les variables aléatoires des quadrilatères afin que ceux-ci se rapprochent d'un rectangle.

Nous pouvons donc choisir les longueurs l_1, l_2 de manière à ce qu'elles soient comprises entre 50 et 200. Pour cela, nous utilisons la fonction *rand()* de *GéoTortue*.
Ainsi : $l_1 = l_2 = 200 - 150 \times rand()$.

Pour que le quadrilatère se rapproche d'un rectangle, nous devons contraindre la longueur l_3 afin qu'elle soit proches de la longueur l_1. Nous pouvons choisir par exemple : $l_3 = l_1 + 20 \times (1 - 2 \times rand())$.

Les angles doivent être proche de l'angle droit. Nous pouvons les choisir afin qu'il soient compris entre 60 et 120, tout en concentrant les valeurs autour de 90°. Ainsi : $\hat{a}_1 = \hat{a}_2 = 90 + 30 \times (-1)^{alea(2)} \times rand()^3$.
En effet, $(-1)^{alea(2)}$ permet d'obtenir aléatoirement les valeurs 1 et -1 et donc $(-1)^{alea(2)} \times rand()^3$ permet d'obtenir des valeurs comprises entre -1 et 1 avec davantage de valeurs proches de 0.

Nous pouvons créer une procédure *quadrilatère_aléatoire* et une procédure *motif* :

```
pour quadrilatere_aleatoire
//variables aléatoires du quadrilatère
l1:=200-150*rand()
l2:=200-150*rand()
l3:=l1+20*((1-2*rand()))
a1:=90+30*(rand()^3)*(-1)^(alea(2))
a2:=90+30*(rand()^3)*(-1)^(alea(2))

quadrilatere l1 l2 l3 a1 a2
fin
```

```
pour motif n
rep n (quadrilatère_aléatoire)
fin
```

Quelques ajustements sont encore nécessaires. En testant la procédure ci-dessus, nous pouvons nous apercevoir que certains quadrilatères sont croisés, ce qui n'est pas le cas dans l'œuvre. Afin d'exclure ce cas, il suffit de conditionner la création du quadrilatère à la contrainte angulaire suivante : $\hat{a}_1 + \hat{a}_2 \leqslant 180$.
Cela nous donne :

```
pour quadrilatere_aleatoire
//variables aléatoires du quadrilatère
l1:=200-150*rand()
l2:=200-150*rand()
l3:=l1+20*((1-2*rand()))
a1:=90+30*(rand()^3)*(-1)^(alea(2))
a2:=90+30*(rand()^3)*(-1)^(alea(2))

si (a1+a2<=180) alors (quadrilatere l1 l2 l3 a1 a2) sinon (quadrilatere_aleatoire)//
quadrilatere non croisé
fin
```

Lorsque l'on teste cette procédure (en éxécutant *motif 20*), nous constatons le parallélisme de certains côtés des quadrilatères *(voir fig. 3)*. Afin de remédier à ce problème, nous pouvons changer légèrement l'orientation de départ de la tortue. Nous avons besoin pour cela d'une variable aléatoire supplémentaire. L'angle qui détermine l'orientation varie de -10° à 10°. Cela nous donne : $a = 10 - 20 \times rand()$.

fig. 3



motif 20 sans changement d'orientation au départ



motif 20 avec changement d'orientation au départ

La procédure *quadrilatère aléatoire* devient :

```
pour quadrilatère_aléatoire
// orientation aléatoire du quadrilatère
a:=10-20*rand()

//variables aléatoires du quadrilatère
l1:=200-150*rand()
l2:=200-150*rand()
l3:=l1+20*((1-2*rand()))
a1:=90+30*(rand()^3)*(-1)^(alea(2))
a2:=90+30*(rand()^3)*(-1)^(alea(2))

si (a1+a2<=180) alors (td a;quadrilatere l1 l2 l3 a1 a2;tg a) sinon (quadrilatere_aleatoire)
fin
```

Choix des couleurs du motif

Dans l'œuvre, il y a deux couleurs. Le rouge est choisi dans l'exemple avec le code *palette RVB(140,0,0)* et le noir avec *palette RVB(0,0,0)*.

Création du tableau

Les différents motifs sont répartis sur une grille de 6 par 6. Plusieurs solutions de programmation sont possibles. Nous pouvons choisir de faire une seule procédure grille en utilisant la fonction *boucle* (correspondant à une boucle *for*) ou bien créer une procédure qui fabriquera une ligne puis une procédure qui fabriquera la grille. C'est la 1ère solution qui est proposée ci-dessous car elle permet de contrôler plus aisément le nombre de quadrilatères composant chaque motif.

Nous utiliserons donc la boucle suivante : boucle m de 1 à 36 (...)

Pour fabriquer la grille, nous choisissons un écart de 200 entre les motifs. Il est nécessaire de déplacer la tortue entre les motifs d'une même ligne et à chaque changement de ligne. Les changements de ligne sont repérés par : $\mathit{reste}(m,6) = 1 \quad \mathit{si}\ m > 1$

Le nombre de quadrilatères composant chaque motif suit une logique croissante *(voir fig. 2)*. Nous pouvons choisir ce nombre en fonction de la position du motif sur la grille (nombre m) de la manière suivante : $\begin{cases} 2+3\times(m-1) & \text{pour la partie noire } (m \leq 18) \\ 2+3\times(36-m) & \text{pour la partie rouge } (m > 18) \end{cases}$

La procédure *grille* est la suivante :

```
pour grille
ecart:=200

boucle m de 1 à 36 (
//changement de ligne
si ((m>1)&&(reste(m,6)==1)) alors (lc;tg 90;av 6*ecart;td 90;av ecart;bc)

//création des lignes
si (m<=18) alors (palette RVB 0 0 0;motif (2+3*(m-1))) sinon (palette RVB 140 0 0;motif (2+3*(36-m)))
lc; td 90; av ecart;tg 90;bc
)

//retour au point de départ
lc;td 90;re 6*ecart;tg 90;re 5*ecart;bc
fin
```

Avec les élèves

Il peut être intéressant que les élèves découvrent par eux-même le théorème d'*Al-Kashi* qui est utilisé dans la réalisation de la pocédure *quadrilatère* (voir les commentaires de l'autre œuvre de Véra MOLNAR appellée elle-aussi *Structure de quadrilatères*).

Un accompagnement particulier doit être réalisé lors de la découverte de la fonction *alea(n)*. Les élèves doivent voir concrètement quel est son effet, par exemple, en éxécutant de nombreuses fois la commande suivante : aff alea(2) et aff $(-1)^{\text{alea}(2)}$.

De la même manière, lors de la mise en place des variables aléatoires, un travail particulier est nécessaire sur l'utilisation de la fonction *rand()* et sur la manière de choisir des nombres dans un intervalle choisi.

Il faudra travailler sur les encadrements et les inégalités en montrant par exemple que :

$$\begin{array}{c} 0 \leqslant rand() \leqslant 1 \\ 0 \leqslant 150\times rand() \leqslant 150 \\ -150 \leqslant -150\times rand() \leqslant 0 \\ 50 \leqslant 200-150\times rand() \leqslant 200 \\ \text{donc } l_1 \in [50;200] \end{array}$$

$$\begin{array}{c} 0 \leqslant rand() \leqslant 1 \\ 0 \leqslant rand()^3 \leqslant 1 \\ -1 \leqslant (-1)^{alea(2)}\times rand()^3 \leqslant 1 \\ -30 \leqslant 30\times(-1)^{alea(2)}\times rand()^3 \leqslant 30 \\ 60 \leqslant 90+30\times(-1)^{alea(2)}\times rand()^3 \leqslant 120 \\ donc\ \hat{a}_1 \in [60;120] \end{array}$$

Ce travail est réalisable au lycée. Il peut être proposé dans le cadre d'un projet et/ou de l'apprentissage de l'algorithmique. Cette activité demande du temps ou tout au moins une certaine habitude de *GéoTortue*.

Il est aussi nécessaire d'accompagner cette activité d'un travail de recherche sur l'artiste, sur le mouvement dans lequel il s'inscrit, sur l'œuvre elle-même (sa date de réalisation, ses dimensions, le support utilisé ...), d'insister sur la différence entre une reproduction à l'ordinateur et le tableau lui-même.

Pour aller plus loin

La figure obtenue avec ce programme n'est pas celle réalisée par Véra Molnar puisque notre construction fait appel à plusieurs paramètres aléatoires. Ainsi, à chaque exécution du programme nous obtenons une figure différente. Pourquoi avoir choisi cette configuration parmi les très nombreuses possibles ? La réponse appartient à l'artiste...

D'autres réalisations sont possibles à partir des procédures écrites pour reproduire cette œuvre. Véra Molnar a réalisé beaucoup d'œuvres à base de quadrilatères quelconques sur d'autres supports que le papier. Une œuvre réalisée sur ordinateur elle aussi nommé *Structure de quadrilatères,* fait l'objet d'une étude similaire.

Sources et liens

- site officiel de Véra Molnar
- site du digital art Museum
- brochure de l'exposition dynamo du Grand Palais (Paris)
- site de la galerie ONIRIS (Rennes)
- site de l'IREM Paris Nord
- site de *GéoTortue*